```
> # Maple worksheet:  Examine the amplitude of the particular solution
  #  to the mass-spring eqn as a function of the forcing frequency.
> # In the equation below, g is for gamma (gamma is a reserved variable)
> # First, define the differential equation:
  Deqn1:=m*diff(u(t),t$2)+g*diff(u(t),t)+k*u(t)=F_0*cos(omega*t);
```

$$Deqn1 := m\left(\frac{d^2}{dt^2}\,u(t)\right) + g\left(\frac{d}{dt}\,u(t)\right) + k\,u(t) = F\_0\cos(\omega\,t) \tag{1}$$

```
> # Next, get the solution:
  U:=rhs(dsolve(Deqn1,u(t)));
```

$$U := e^{\frac{1}{2}\frac{\left(-g+\sqrt{g^2-4\,k\,m}\right)t}{m}}\,\_C2 + e^{-\frac{1}{2}\frac{\left(g+\sqrt{g^2-4\,k\,m}\right)t}{m}}\,\_C1 + \frac{\left(\left(-m\,\omega^2+k\right)\cos(\omega\,t)+\sin(\omega\,t)\,g\,\omega\right)F\_0}{m^2\,\omega^4+\left(g^2-2\,k\,m\right)\omega^2+k^2} \tag{2}$$

```
> # Find the amplitude of the particular part of the solution.
  A:=coeff(U,cos(omega*t)); B:=coeff(U,sin(omega*t));
```

$$A := \frac{\left(-m\,\omega^2+k\right)F\_0}{m^2\,\omega^4+\left(g^2-2\,k\,m\right)\omega^2+k^2}$$

$$B := \frac{g\,\omega\,F\_0}{m^2\,\omega^4+\left(g^2-2\,k\,m\right)\omega^2+k^2} \tag{3}$$

```
> R:=simplify(sqrt(A^2+B^2));
```

$$R := \sqrt{\frac{F\_0^2}{m^2\,\omega^4+g^2\,\omega^2-2\,k\,m\,\omega^2+k^2}} \tag{4}$$

```
> #Find the max value of R (using omega as the variable)
> Ans1:=solve(diff(R,omega)=0, omega);
```

$$Ans1 := 0,\ \frac{1}{2}\,\frac{\sqrt{-2\,g^2+4\,k\,m}}{m},\ -\frac{1}{2}\,\frac{\sqrt{-2\,g^2+4\,k\,m}}{m} \tag{5}$$

```
> MaxOmega:=Ans1[2];
```

$$MaxOmega := \frac{1}{2}\,\frac{\sqrt{-2\,g^2+4\,k\,m}}{m} \tag{6}$$

```
> Rmax:=simplify(subs(omega=MaxOmega,R));
```

$$Rmax := 2\sqrt{-\frac{F\_0^2\,m^2}{\left(g^2-4\,k\,m\right)g^2}} \tag{7}$$

```
>  # Example:  Plot the amplitude R as a function of omega if F_0=
  1, m=1, k=4 and g decreases
> Rf0:=subs(m=1,k=4,g=1/2,F_0=1,R);
```

$$Rf0 := \sqrt{\frac{1}{\omega^4 - \frac{31}{4}\omega^2 + 16}} \quad \textbf{(8)}$$

```
> Rf1:=subs(m=1,k=4,g=1/5,F_0=1,R);
```

$$Rf1 := \sqrt{\frac{1}{\omega^4 - \frac{199}{25}\omega^2 + 16}} \quad \textbf{(9)}$$

```
> Rf2:=subs(m=1,k=4,g=1/10,F_0=1,R);
```

$$Rf2 := \sqrt{\frac{1}{\omega^4 - \frac{799}{100}\omega^2 + 16}} \quad \textbf{(10)}$$

```
> plot({Rf0,Rf1,Rf2},omega=0..4);
```




```
> # The three graphs differ only in the amount of damping that is
  applied.
  #  The bottom graph has the most damping, and the top graph has
  the least.
> # This shows two things:  (1) For any single curve in the graph,
```

```
  the maximum
> #  amplitude is approximately 2 (the natural frequency), and (2)
  as the damping
> #  decreases the maximum amplitude increases...
```